
A TAX RECEIPTS MONITORING AND FORECASTING MODEL FOR LITHUANIA 

By William B. Trautman1 

OVERVIEW 

This paper presents a general tax receipts monitoring and forecasting model and shows 

how it has been applied to Lithuania.  The purpose of the model is to generate receipts 

forecasts as part of the annual budget process and to monitor receipts over the course of 

the year relative to amounts projected in the budget.  The model uses actual monthly re-

ceipts data and forecasts receipts for the current and subsequent years based on expected 

changes in real GDP, the price level, tax rates, and any behavioral effects that may be as-

sociated with changes in tax rates.  Since the model is based on monthly receipts data, it 

is possible to update the forecasts on a monthly basis and therefore monitor receipts over 

the course of the year relative to amounts projected in the budget.  The ability to update 

the forecasts on a monthly basis provides budget officials with a tool for identifying po-

tential receipts shortfalls before they become a serious problem. 

The model is particularly applicable to economies in transition, because it requires 

only a minimal amount of relatively current data and few assumptions.  The model is de-

signed so that those with a limited technical understanding of forecasting or computers 

can input data, adjust parameters, run the model, and interpret the results.  The model 

therefore represents a useful tool which can be integrated quickly and easily into a budget 

or tax analysis department.  The model also establishes a foundation for the development 

                                                 
1 William B. Trautman is a Development Associate at the Harvard Institute for International Development 
(HIID) and a Research Fellow in the International Tax Program at the Harvard Law School.  The model 
was initially developed while the author was the U.S. Treasury Department’s Tax Policy Advisor to the 
Slovak Ministry of Finance.  It has since been adapted for a number of other countries, including Lithuania 
as part of HIID’s CAER Lithuania project.  The views expressed in this paper do not necessarily reflect 
those of the institutions with which the author has been associated. 



 

 

 

of more precise forecasting techniques over time.  Indeed, as analysts become increas-

ingly familiar with the model, it is hoped that they would refine the model by incorporat-

ing more detailed data on the structure of the receipts sources being modeled and macro-

economic measures that more closely proxy the underlying tax bases.  A description of 

the way the model was implemented in Microsoft Access is presented in Attachment A.  

The input and output forms for the model developed for Lithuania are presented in At-

tachments 1 to 3. 

THE GENERAL MODEL 

The model presented in this paper is quite simple.  The forecast of receipts for any budget 

period is equal to actual receipts in the subperiods for which receipts data are available 

plus projected receipts in the subperiods for which the data are not available.  The pro-

jected receipts, in turn, are equal to the actual or projected receipts in the comparable 

subperiod of the previous budget period, adjusted by three factors: 1) the expected change 

in the tax base holding the tax rate constant, 2) the change in the tax rate, and 3) the 

change in the tax base associated with the change in the tax rate.  The first factor is a 

weighted average of: 1) the expected change in GDP, and 2) the ratio of actual receipts to 

date in the current budget period to actual receipts in the comparable subperiods of the 

previous budget period.  The weight varies from 0 to 1 depending on the proportion of 

subperiods for which actual receipts data are available. 

The general form of the model is as follows: 

 yyyT FPEA ⋅+⋅= TT  

and i indexes each of the n subperiods in budget period y.  In Lithuania, as well as in the 

other countries for which this model has been adapted, the relevant budget periods are 

calendar years, and the subperiods are months. 

If ei y−
=

1

budget period y are estimated as follows: 

where τ i
 is the tax rate applicable to subperiod i, δ is a growth factor, and η is the user-

defined elasticity of the tax base with respect to changes in the tax rate.  If 0

The growth factor, δ, is a weighted average of: 1) the expected change in GDP, 

and 2) the ratio of actual receipts to date in budget period y to actual receipts in the com-

parable subperiods of the previous budget period y-1, as follows: 

where GDP is the actual or projected gross domestic product.2 

The weight, α, varies from 0 when receipts data are available for all subperiods in 

the current budget period to 1 when no receipts data are available for any subperiod in the 

current budget period, as follows: 

MODEL CHARACTERISTICS 

The model presented in this paper differs in a number of important respects from more 

typical receipts forecasting models, an example of which is a model developed for Kenya 

as part of the IMF Institute Financial Policy Workshops.3  These more typical models are 

often based on an estimated relationship between annual tax receipts and a proxy for the 

tax base.  Based on this estimated relationship, forecasts of the proxy variable are used to 

generate receipts forecasts over the relevant budget period.  The models are based on 

variables that proxy (i.e., are correlated with) the tax base, because forecasts of the actual 

tax base often are not available. 

                                                 
2 The model generally uses nominal GDP as a proxy for the tax base.  It uses real GDP in the case of per 
unit excise taxes, however, because their tax bases are assumed to be invariant with respect to changes in 
the price level. 



 

 

 

One important difference between the models is the scope of the data on which 

they are based.  The Lithuania model is based on the most current twelve months of re-

ceipts data that are available, whereas the Kenya model is based on annual receipts data 

that may go back over many years.  Since the forecasts of the Lithuania model are based 

on actual receipts in a single time period (i.e., subperiod i of budget period y-1), they are 

more susceptible to random fluctuations in receipts than the Kenya model which effec-

tively is based on receipts averaged over multiple time periods.  The forecasts of the 

Lithuania model, however, are less susceptible to structural changes in the relationship 

between tax receipts and the proxy variable over the period of the data, because they are 

based on more current data.  Since these structural changes may reflect underlying 

changes in the economy, the Lithuania model may be more appropriate for countries ex-

periencing significant economic change (e.g., developing or transition economies). 

The forecasts of both models depend, in part, on forecasts of the proxy for the tax 

base (e.g., nominal GDP), presumably from a macroeconomic model of the economy.  

The forecasts of the Kenya model are based on statistically estimated relationships be-

tween tax receipts and the proxy for the tax base.  The precision of these forecasts is 

questionable, given the limited amount of data used to derive the estimates, and given the 

possibility of changes in the structural relationship over the period of the underlying data.  

The forecasts of the Lithuania model are based on a growth factor, which is a weighted 

average of: 1) the expected change in the proxy variable, and 2) the change in actual re-

ceipts to date in the current year relative to receipts in the same months of the previous 

year.  While this growth factor may not be entirely satisfactory from a theoretical stand-

                                                                                                                                                 
3 See IMF Institute Financial Policy Workshops: The Case of Kenya (1991). 



 

 

 

point,4 it attempts to reduce any inaccuracies in the forecasts of the proxy variable by in-

jecting empirical information (i.e., actual receipts) on the change in the size of the tax 

base in the relevant year. 

Both models attempt to account for the discretionary changes in the tax system af-

fecting receipts.  The Kenya model makes ad hoc adjustments to the time series of tax 

receipts in order to ensure that the time series is consistent with current law.  In turn, this 

ensures that the estimate of the relationship between tax receipts and the proxy variable 

(and therefore the forecast) is unbiased.  In theory, the Kenya model accounts for all dis-

cretionary changes affecting receipts but, for that reason, cannot be explicit about how 

the changes affect receipts.  The Lithuania model is somewhat more limited in scope in 

that it only accounts for changes in tax rates, but it is explicit about how those changes 

affect receipts.  The model can therefore be used to estimate the revenue effects of pro-

posed changes in tax rates. 

CONCLUSION 

This paper has presented a general model for monitoring and forecasting tax receipts and 

shows how it has been applied to Lithuania.  The model generates forecasts of receipts as 

part of the regular budget process and monitors receipts over the course of the year rela-

tive to amounts projected in the budget.  Since the model is based on actual monthly re-

ceipts data, it can be updated on a monthly basis and used as a tool for identifying poten-

tial receipts shortfalls before they become a serious problem.  The model is designed so 

                                                 
4 The difficulty arises from the inconsistency between: 1) the time periods for which the receipts projec-
tions are made, and 2) the time periods associated with the growth factor.  In particular, the model makes 
projections only for those months for which actual receipts data are not available.  The growth factor, in 
contrast, is based on the expected change in GDP for the entire year and on actual receipts for those months 



 

 

 

that those with a limited understanding of modeling or computers can input data, adjust 

parameters, run the model, and interpret the results.  The model therefore represents a 

useful tool that can be integrated quickly and easily into a budget or tax analysis depart-

ment.  It can also be used as a tool for educating users about receipts forecasting and 

revenue estimating.  Indeed, as users become increasingly familiar with the model, they 

can modify the forecasting algorithm in ways that better reflect both the available data 

and the structure of the receipts sources being modeled. 

                                                                                                                                                 
for which actual receipts data are available. 



 

 

 

APPENDIXA 

MODEL IMPLEMENTATION IN MICROSOFT ACCESS 

The tax receipts monitoring and forecasting model for Lithuania is implemented through 
a series of tables, forms, reports, macros, and modules in Microsoft Access. 

TABLES 

The model is based on data contained in the following three tables: 

Assumptions.  This table contains a series of variables representing the elasticities associ-
ated with each tax, a series of variables indicating whether or not the tax is imposed on a 
per unit basis, and variables representing nominal GDP and the price level for the current 
and subsequent years. 

Receipts Actual.  This table contains a series of variables representing the actual monthly 
receipts for each tax and a series of variables representing the monthly tax rate index for 
each tax.  It also contains variables for the year and month of the data.  The table has 72 
records representing the 72 months between January 1995 and December 2000. 

Receipts Projected.  This table contains a series of variables representing the monthly re-
ceipts projected by the model for each tax in the current and subsequent years and a series 
of variables containing the amounts in the budget for the current year.  It also contains 
variables identifying the tax and the current and subsequent years.  The table has 17 re-
cords representing the 17 different taxes. 

FORMS 

The model uses the following five forms to handle input and output: 

Assumptions.  This form allows the analyst to modify the data in the Assumptions table, 
including the elasticities associated with each tax, whether or not each tax is imposed on 
a per unit basis, nominal GDP, and the price level. 

Receipts Actual.  This form allows the analyst to modify the data in the Receipts Actual 
table, including the monthly tax receipts and the monthly tax rate index. 

Receipts Projected CY.  This form presents the forecast of monthly tax receipts associ-
ated with each tax for the current year. 

Receipts Projected NY.  This form presents the forecast of monthly tax receipts associ-
ated with each tax for the next year. 

Receipts Projected Summary.  This form presents the forecast of annual tax receipts asso-
ciated with each tax for the current and subsequent years.  It also allows the analyst to 
input tax receipts projections from the budget and presents the projected surplus or defi-
cit. 



 

 

 

REPORTS 

The model uses the following reports to print out information from the corresponding 
forms: Assumptions, Receipts Projected CY, Receipts Projected NY, and Receipts Sum-
mary. 

MACROS 

The model uses a series of macros to perform certain functions, including opening forms, 
printing reports, updating the data, and running the model.  The macros Actual, Assump-
tions, Projected CY, Projected NY, and Summary open the corresponding forms.  The 
macro Print prints the Summary, Assumptions, Receipts Projected CY, and Receipts Pro-
jected NY forms.  These macros run when corresponding buttons are pressed in different 
forms.  The Update Model macro copies data to the appropriate table and runs the model 
every time the data are updated on the Assumptions and Receipts Actual forms. 

MODULE 

The module Model contains the computer code that executes the calculations described in 
the body of this paper to create the forecasts.  The annotated computer code is presented 
in the following pages. 



 

Option Compare Database 
Option Explicit 
 
' Dimension Database and Tables 
 
Dim Db As Database 
 
Dim TbA As Recordset 
Dim TbS As Recordset 
Dim TbP As Recordset 
 
' Define Constants for Array Dimensions 
 
Const TAXES = 17 
Const MONTHS = 72 
 
Const RCPT = 1 
Const RATE = 2 
Const ELAS = 1 
Const UNIT = 2 
Const CY = 1 
Const NY = 2 
 
' Dimension Arrays 
 
Dim Data(1 To TAXES, 1 To MONTHS, RCPT To RATE) As Single 
Dim Assume(1 To TAXES, ELAS To UNIT) As Single 
 
Dim Nom_GDP(CY To NY) As Single 
Dim Real_GDP(CY To NY) As Single 
 
Dim Receipts(1 To MONTHS) As Single 
 
Dim YTD_PY(1 To TAXES) As Single 
Dim YTD_CY(1 To TAXES) As Single 
 
' Dimension Indexes 
 
Dim m As Integer 
Dim t As Integer 
Dim y As Integer 
 
' Dimension Other Variables 
 
Dim Record As Integer 
Dim Alpha As Single 



 

 

 

Function max(x As Single, y As Single) As Single 
   
  If x > y Then 
    max = x 
  Else 
    max = y 
  End If 
   
End Function 
Function Forecast() 
 
Set Db = CurrentDb() 
 
Set TbA = Db.OpenRecordset("Receipts Actual", dbOpenTable) 
Set TbS = Db.OpenRecordset("Assumptions", dbOpenTable) 
Set TbP = Db.OpenRecordset("Receipts Projected", dbOpenTable) 
 
' Initialize Data Array (Tax * Month * Receipts and Tax Rate Index) 
 
TbA.MoveFirst 
 
For m = 1 To MONTHS 
   
  Data(1, m, RCPT) = TbA!Personal_Income_Tax 
  Data(2, m, RCPT) = TbA!Profit_Tax 
  Data(3, m, RCPT) = TbA!Property_Tax 
  Data(4, m, RCPT) = TbA!Value_Added_Tax 
  Data(5, m, RCPT) = TbA!Excise_Taxes 
  Data(6, m, RCPT) = TbA!Customs_Duties 
  Data(7, m, RCPT) = TbA!State_Fees_and_Charges 
  Data(8, m, RCPT) = TbA!Consul_Tax 
  Data(9, m, RCPT) = TbA!Taxes_on_Market_Places 
  Data(10, m, RCPT) = TbA!Bank_of_Lithuania_Profit 
  Data(11, m, RCPT) = TbA!Natural_Resource_Tax 
  Data(12, m, RCPT) = TbA!Tax_on_Fixed_State_Assets 
  Data(13, m, RCPT) = TbA!Interest_on_Loans 
  Data(14, m, RCPT) = TbA!Stock_Dividends 
  Data(15, m, RCPT) = TbA!Penalty_Tax 
  Data(16, m, RCPT) = TbA!Pollution_Taxes 
  Data(17, m, RCPT) = TbA!Other_Non_Tax_Revenue 
   
  Data(1, m, RATE) = TbA!Personal_Income_Tax_TRI 
  Data(2, m, RATE) = TbA!Profit_Tax_TRI 
  Data(3, m, RATE) = TbA!Property_Tax_TRI 
  Data(4, m, RATE) = TbA!Value_Added_Tax_TRI 
  Data(5, m, RATE) = TbA!Excise_Taxes_TRI 



 

 

 

  Data(6, m, RATE) = TbA!Customs_Duties_TRI 
  Data(7, m, RATE) = TbA!State_Fees_and_Charges_TRI 
  Data(8, m, RATE) = TbA!Consul_Tax_TRI 
  Data(9, m, RATE) = TbA!Taxes_on_Market_Places_TRI 
  Data(10, m, RATE) = TbA!Bank_of_Lithuania_Profit_TRI 
  Data(11, m, RATE) = TbA!Natural_Resource_Tax_TRI 
  Data(12, m, RATE) = TbA!Tax_on_Fixed_State_Assets_TRI 
  Data(13, m, RATE) = TbA!Interest_on_Loans_TRI 
  Data(14, m, RATE) = TbA!Stock_Dividends_TRI 
  Data(15, m, RATE) = TbA!Penalty_Tax_TRI 
  Data(16, m, RATE) = TbA!Pollution_Taxes_TRI 
  Data(17, m, RATE) = TbA!Other_Non_Tax_Revenue_TRI 
     
  TbA.MoveNext 
Next 
   
' Initialize Assume Array (Tax * Elasticity and Unit Tax Dummy) 
   
TbS.MoveFirst 
   
Assume(1, ELAS) = TbS!Personal_Income_Tax 
Assume(2, ELAS) = TbS!Profit_Tax 
Assume(3, ELAS) = TbS!Property_Tax 
Assume(4, ELAS) = TbS!Value_Added_Tax 
Assume(5, ELAS) = TbS!Excise_Taxes 
Assume(6, ELAS) = TbS!Customs_Duties 
Assume(7, ELAS) = TbS!State_Fees_and_Charges 
Assume(8, ELAS) = TbS!Consul_Tax 
Assume(9, ELAS) = TbS!Taxes_on_Market_Places 
Assume(10, ELAS) = TbS!Bank_of_Lithuania_Profit 
Assume(11, ELAS) = TbS!Natural_Resource_Tax 
Assume(12, ELAS) = TbS!Tax_on_Fixed_State_Assets 
Assume(13, ELAS) = TbS!Interest_on_Loans 
Assume(14, ELAS) = TbS!Stock_Dividends 
Assume(15, ELAS) = TbS!Penalty_Tax 
Assume(16, ELAS) = TbS!Pollution_Taxes 
Assume(17, ELAS) = TbS!Other_Non_Tax_Revenue 
     
Assume(1, UNIT) = TbS!Personal_Income_Tax_IT 
Assume(2, UNIT) = TbS!Profit_Tax_IT 
Assume(3, UNIT) = TbS!Property_Tax_IT 
Assume(4, UNIT) = TbS!Value_Added_Tax_IT 
Assume(5, UNIT) = TbS!Excise_Taxes_IT 
Assume(6, UNIT) = TbS!Customs_Duties_IT 
Assume(7, UNIT) = TbS!State_Fees_and_Charges_IT 
Assume(8, UNIT) = TbS!Consul_Tax_IT 



 

 

 

Assume(9, UNIT) = TbS!Taxes_on_Market_Places_IT 
Assume(10, UNIT) = TbS!Bank_of_Lithuania_Profit_IT 
Assume(11, UNIT) = TbS!Natural_Resource_Tax_IT 
Assume(12, UNIT) = TbS!Tax_on_Fixed_State_Assets_IT 
Assume(13, UNIT) = TbS!Interest_on_Loans_IT 
Assume(14, UNIT) = TbS!Stock_Dividends_IT 
Assume(15, UNIT) = TbS!Penalty_Tax_IT 
Assume(16, UNIT) = TbS!Pollution_Taxes_IT 
Assume(17, UNIT) = TbS!Other_Non_Tax_Revenue_IT 
     
' Create Nominal and Real GDP Arrays (GDP * Current Year and Next Year) 
     
Nom_GDP(CY) = 1 + TbS!Nom_GDP_CY 
Nom_GDP(NY) = 1 + TbS!Nom_GDP_NY 
 
Real_GDP(CY) = (1 + TbS!Nom_GDP_CY) / (1 + TbS!Price_Level_CY) 
Real_GDP(NY) = (1 + TbS!Nom_GDP_NY) / (1 + TbS!Price_Level_NY) 
   
' Extend Most Recent Tax Rate Index into Future if Equal to Zero 
   
For t = 1 To TAXES 
  For m = 1 To MONTHS 
    If Data(t, m, RATE) = 0 Then Data(t, m, RATE) = Data(t, m - 1, RATE) 
  Next 
Next 
 
' Create Receipts Array (Total Receipts By Month) 
 
For m = 1 To MONTHS 
  Receipts(m) = 0 
Next 
 
For m = 1 To MONTHS 
  For t = 1 To TAXES 
    Receipts(m) = Receipts(m) + Data(t, m, RCPT) 
  Next 
Next 
 
' Calculates Record (Number of Months for which Receipts Data Exist) 
 
Record = 0 
 
For m = 1 To MONTHS 
  If Receipts(m) <> 0 Then 
    Record = Record + 1 
  End If 



 

 

 

Next 
 
' Calculates Alpha (Weight on GDP in Growth Factor) 
 
Alpha = 1 - (Record - Int(Record / 12) * 12) / 12 
 
' Calculates Year-To-Date Receipts for Previous and Current Years 
 
For t = 1 To TAXES 
  YTD_PY(t) = 0 
  YTD_CY(t) = 0 
Next 
 
For t = 1 To TAXES 
  For m = Int((Record - 12) / 12) * 12 + 1 To Record - 12 
    YTD_PY(t) = YTD_PY(t) + Data(t, m, RCPT) 
  Next 
Next 
 
For t = 1 To TAXES 
  For m = Int(Record / 12) * 12 + 1 To Record 
    YTD_CY(t) = YTD_CY(t) + Data(t, m, RCPT) 
  Next 
Next 
 
' Forecast Receipts for Current Year 
 
For t = 1 To TAXES 
  For m = Record + 1 To Int(Record / 12) * 12 + 12 
     
    ' Unit Taxes 
     
    If Assume(t, UNIT) = -1 Then 
      Data(t, m, RCPT) = Data(t, m - 12, RCPT) _ 
          * (Alpha * Real_GDP(CY) + (1 - Alpha) * YTD_CY(t) / max(YTD_PY(t), 1)) _ 
          * ((Data(t, m, RATE) / Data(t, m - 12, RATE)) ^ (1 + Assume(t, ELAS))) 
     
    ' Other Taxes 
     
    Else 
      Data(t, m, RCPT) = Data(t, m - 12, RCPT) _ 
          * (Alpha * Nom_GDP(CY) + (1 - Alpha) * YTD_CY(t) / max(YTD_PY(t), 1)) _ 
          * ((Data(t, m, RATE) / Data(t, m - 12, RATE)) ^ (1 + Assume(t, ELAS))) 
    End If 
  Next 
Next 



 

 

 

 
' Forecast Receipts for Next Year 
 
For t = 1 To TAXES 
  For m = Int(Record / 12) * 12 + 13 To Int(Record / 12) * 12 + 24 
     
    ' Unit Taxes 
     
    If Assume(t, UNIT) = -1 Then 
      Data(t, m, RCPT) = Data(t, m - 12, RCPT) _ 
          * Real_GDP(NY) _ 
          * ((Data(t, m, RATE) / Data(t, m - 12, RATE)) ^ (1 + Assume(t, ELAS))) 
     
    ' Other Taxes 
     
    Else 
      Data(t, m, RCPT) = Data(t, m - 12, RCPT) _ 
          * Nom_GDP(NY) _ 
          * ((Data(t, m, RATE) / Data(t, m - 12, RATE)) ^ (1 + Assume(t, ELAS))) 
    End If 
  Next 
Next 
   
' Copy Forecast to Projected Receipts Table 
   
TbP.MoveFirst 
   
For t = 1 To TAXES 
  TbP.Edit 
  TbP!CY = Int(Record / 12) + 1995 
  TbP!NY = TbP!CY + 1 
  TbP!January_CY = Data(t, Int(Record / 12) * 12 + 1, RCPT) 
  TbP!February_CY = Data(t, Int(Record / 12) * 12 + 2, RCPT) 
  TbP!March_CY = Data(t, Int(Record / 12) * 12 + 3, RCPT) 
  TbP!April_CY = Data(t, Int(Record / 12) * 12 + 4, RCPT) 
  TbP!May_CY = Data(t, Int(Record / 12) * 12 + 5, RCPT) 
  TbP!June_CY = Data(t, Int(Record / 12) * 12 + 6, RCPT) 
  TbP!July_CY = Data(t, Int(Record / 12) * 12 + 7, RCPT) 
  TbP!August_CY = Data(t, Int(Record / 12) * 12 + 8, RCPT) 
  TbP!September_CY = Data(t, Int(Record / 12) * 12 + 9, RCPT) 
  TbP!October_CY = Data(t, Int(Record / 12) * 12 + 10, RCPT) 
  TbP!November_CY = Data(t, Int(Record / 12) * 12 + 11, RCPT) 
  TbP!December_CY = Data(t, Int(Record / 12) * 12 + 12, RCPT) 
    
  TbP!January_NY = Data(t, Int(Record / 12) * 12 + 13, RCPT) 
  TbP!February_NY = Data(t, Int(Record / 12) * 12 + 14, RCPT) 



 

 

 

  TbP!March_NY = Data(t, Int(Record / 12) * 12 + 15, RCPT) 
  TbP!April_NY = Data(t, Int(Record / 12) * 12 + 16, RCPT) 
  TbP!May_NY = Data(t, Int(Record / 12) * 12 + 17, RCPT) 
  TbP!June_NY = Data(t, Int(Record / 12) * 12 + 18, RCPT) 
  TbP!July_NY = Data(t, Int(Record / 12) * 12 + 19, RCPT) 
  TbP!August_NY = Data(t, Int(Record / 12) * 12 + 20, RCPT) 
  TbP!September_NY = Data(t, Int(Record / 12) * 12 + 21, RCPT) 
  TbP!October_NY = Data(t, Int(Record / 12) * 12 + 22, RCPT) 
  TbP!November_NY = Data(t, Int(Record / 12) * 12 + 23, RCPT) 
  TbP!December_NY = Data(t, Int(Record / 12) * 12 + 24, RCPT) 
  TbP.Update 
  TbP.MoveNext 
Next 
 
TbA.Close 
TbS.Close 
TbP.Close 
 
End Function 

1995January

Personal Income Tax 110,093

Profit Tax 21,259

Property Tax 1,124

Land Tax 0

Land-Rent Tax 0

Wealth Tax 0

Value-Added Tax 129,636

Excise Taxes 42,330

Customs Duties 9,487

Road Tax 14,447

Description
Budgeted

Summary of Receipts Projections
1999

Projected Surplus Projected

2000

Attachment 3
A Receipts Monitoring and Forecasting Model for Lithuania

Personal Income Tax 2,627,414 2,627,4142,794,600 -167,186

Profit Tax 397,661 397,661525,640 -127,979

Property Tax 246,687 246,687239,691 6,996

Value-Added Tax 3,544,126 3,544,1264,266,212 -722,086

Excise Taxes 1,309,502 1,309,5021,555,000 -245,498

Customs Duties 190,309 190,309251,250 -60,941

State Fees and Charges 196,526 196,526200,000 -3,474

Consul Tax 14,944 14,94435,000 -20,056

Taxes on Market Places 7,849 7,8495,922 1,927

Bank of Lithuania Profit 60,252 60,25230,000 30,252

Natural Resource Tax 30,342 30,34252,188 -21,846

Tax on Fixed State Assets 7,832 7,83212,637 -4,805

Interest on Loans 100,783 100,783137,133 -36,350

Stock Dividends 26,221 26,22156,000 -29,779

Penalty Tax 56,207 56,20753,000 3,207

Pollution Taxes 22,120 22,12017,627 4,493

Other Non-Tax Revenue 132,278 132,27860,000 72,278

Total 8,971,052 8,971,05210,291,900 -1,320,848

Actual Receipts Projections for 
Current Year

Projections for 
Next YearAssumptions Print


